
Riciclare un subfile?

chiamata di un 
programma

chiamata di una 
stored procedure SQL



Result set

 E' possibile creare un programma RPG
pubblicato come stored procedure SQL
che restituisce un resultset partendo
da un cursore SQL o da una struttura
dati array

 dove restituire il result set:

 to client: la procedura più esterna nello stack

 to caller: la procedura precedente nello stack
delle chiamate (default)

Come restituire un result set?

 Ci sono due modi per restituire un result set da un programma 

embedded SQL RPG

 result set array

 result set cursor

Restituire un result set array/1

1. Dichiarazione di una 
struttura dati array e una 
variabile per il conteggio 
degli elementi da restituire

Power coffee - MK1

6

RESSET1.SQLRPGLE

dcl-ds resultSet dim(5) qualified;
Codice char(6);
Cognome varchar(30);
Stipendio packed(13:2);

end-ds;
dcl-s rowCount int(10);

2. Valorizzare i campi della 
struttura dati per il numero 
di elementi desiderati e la 
variabile rowCount con il 
numero totale degli elementi

3. Restituire l'array come result
set

exec sql
set result sets array :resultSet
for :rowCount rows;

Non usare actgrp(*new)
altrimenti al client SQL non viene 
restituito il result set



Restituire un result set array/2

4. Creare una stored procedure SQL che esegue il programma RPG

create or replace procedure RESSET1()
dynamic result sets 1
language rpgle
specific RESSET1
external name RESSET1
parameter style general;

label on procedure RESSET1
is 'Result set - esempio da array';

RESSET1.SQL

Numero di 
result set 
restituiti

Nome del 
programma 

RPG

modalità passaggio 
parametri tra stored

procedure e RPG



Restituire un result set da un cursore SQL

1. Dichiarare e aprire il 
cursore

exec sql
declare cEmp cursor for
select EMPNO, LASTNAME, WORKDEPT, DEPTNAME
from EMPLOYEE
inner join DEPARTMENT
on WORKDEPT = DEPTNO

order by WORKDEPT, LASTNAME;

exec sql
open cEmp;exec sql

set result sets cursor cEmp;

Non bisogna chiudere il cursore 
altrimenti il result set non viene 
restituito

2. Restituire il result
set

3. Come nell'esempio precedente creare 
una stored procedure SQL che esegue 
il programma RPG



Ricevere dei parametri

 La stored procedure SQL e di conseguenza anche il programma 
RPG che richiama possono ricevere uno o più parametri in input

create or replace procedure RESSET3(Department char(3))
dynamic result sets 1
language rpgle
specific RESSET3
external name RESSET3
parameter style general;

dcl-pr RESSET3 extpgm('RESSET3');
*n char(3);

end-pr;
dcl-pi RESSET3;
pmDept char(3);

end-pi;

Accedere a un result set da RPG/1

 da 7.1 una stored procedure può restituire un result set anche se 

chiamata in una istruzione embedded SQL all'interno di un 

programma RPG

 è un'operazione molto facilmente

 prima della 7.1 occorreva utilizzare SQL CLI e occorreva più "fatica"

Accedere a un result set da RPG/2
1. si utilizza un result set locator, ovvero un puntatore al result set

2. si esegue la stored procedure

3. si associa il result set locator alla procedura

4. si alloca un cursore per leggere il result set associato al locator

dcl-s MyResult sqltype(result_set_locator);

exec sql call RESSET3(:pmDept);

exec sql
associate result set locator(:MyResult)
with procedure RESSET3;

exec sql
allocate cResult cursor
for result set :MyResult;

non è necessario
aprire il cursore, si
può leggerlo
direttamente con 
fetch



Come riciclare un subfile?

 Le modifiche da apportare a un programma che gestisce un subfile
affinché possa diventare una stored procedure che restituisce un result
set sono molto limitate

1. Identificare il tipo di job: se interattivo viene emesso il subfile, altrimenti 
se batch non si apre il file video e si restituisce il result set

2. Aggiungere dei parametri di input facoltativi equivalenti ai campi nel 
video che servono per filtrare il subfile

3. Condizionare apertura e le operazioni di I/O sul file video in base al tipo 
di job

4. Se i dati esposti nel subfile derivano dalla lettura di un cursore SQL si 
può restituire direttamente il cursore; altrimenti nella subroutine di 
scrittura del record di subfile si valorizza la struttura dati array

5. Prima della fine del programma restituire il result set con set result
sets

/* ESEMPIO PROGRAMMA RPG USATO IN STORED PROCEDURE
   PER RESTITUIRE UN RESULT SET

   (c) 2022 MarkOneTools - www. markonetools.it */

create or replace procedure RESSET1()
  dynamic result sets 1
  language rpgle
  specific RESSET1
  external name RESSET1
  parameter style general;

label on procedure RESSET1 is 'Result set - esempio da array';


**free

// ESEMPIO PROGRAMMA RPG USATO IN STORED PROCEDURE
//  PER RESTITUIRE UN RESULT SET
//  (c) 2022 MarkOneTools - www. markonetools.it

dcl-s rowCount int(10);
dcl-ds resultSet dim(5) qualified;
  Codice char(6);
  Cognome varchar(30);
  Stipendio packed(13:2);
end-ds;

resultSet(1).Codice = 'WAM';
resultSet(1).Cognome   = 'Mozart';
resultSet(1).Stipendio = 4000.00;

resultSet(2).Codice = 'GV';
resultSet(2).Cognome   = 'Verdi';
resultSet(2).Stipendio = 7000.00;

resultSet(3).Codice = 'LB';
resultSet(3).Cognome   = 'Beethoven';
resultSet(3).Stipendio = 10000.00;

rowCount = 3;

exec sql
  set result sets array :resultSet
    for :rowCount rows;

*inlr = *on;
return;


**free

// ESEMPIO UTILIZZO STORED PROCEDURE CHE RESTITUISCE UN RESULT SET

// (c) 2022 MarkOneTools - www.markonetools.it

ctl-opt copyright('MarkOneTools')
        decedit('0,')
        indent(' ')
        option(*nodebugio: *srcstmt: *showcpy)
        expropts(*resdecpos)
        extbinint(*yes);

// variabile di tipo SQL che rappresenta un puntatore al result set
dcl-s MyResult sqltype(result_set_locator);

// output lettura cursore
dcl-ds dsOut inz;
  Codice char(6);
  Cognome varchar(30);
  Stipendio packed(13:2);
end-ds;

// esecuzione stored procedure
exec sql
  call RESSET1();

// associazione tra la variabile result set e la stored procedure
exec sql
  associate result set locator(:MyResult)
    with procedure RESSET1;

// associa la variabile result set a un cursore
exec sql
  allocate cResult cursor for
    result set :MyResult;

dow *on;
  exec sql
    fetch next from cResult
      into :dsOut;

  if sqlcode = 100;
    leave;
  endif;

  // ... elaborazione record...
  // ... bla bla bla ...

enddo;

exec sql
  close cResult;

*inlr = *on;
return;


/* ESEMPIO PROGRAMMA RPG USATO IN STORED PROCEDURE
   PER RESTITUIRE UN RESULT SET

   (c) 2022 MarkOneTools - www. markonetools.it */

create or replace procedure RESSET2()
  dynamic result sets 1
  language rpgle
  specific RESSET2
  external name RESSET2
  parameter style general;

label on procedure RESSET2 is 'Result set - esempio da cursore';


**free

// ESEMPIO PROGRAMMA RPG USATO IN STORED PROCEDURE
//  PER RESTITUIRE UN RESULT SET PARTENDO DA UN CURSORE
//  (c) 2022 MarkOneTools - www. markonetools.it
ctl-opt copyright('MarkOneTools')
        decedit('0,')
        indent(' ')
        option(*nodebugio: *srcstmt: *showcpy)
        expropts(*resdecpos)
        extbinint(*yes);

exec sql
  declare cEmp cursor for
    select EMPNO, LASTNAME, WORKDEPT, DEPTNAME
      from EMPLOYEE
        inner join DEPARTMENT
          on WORKDEPT = DEPTNO
      order by WORKDEPT, LASTNAME;

exec sql
  open cEmp;

exec sql
  set result sets cursor cEmp;

*inlr = *on;
return;


**free

// ESEMPIO UTILIZZO STORED PROCEDURE CHE RESTITUISCE UN RESULT SET

// (c) 2022 MarkOneTools - www.markonetools.it

ctl-opt copyright('MarkOneTools')
        decedit('0,')
        indent(' ')
        option(*nodebugio: *srcstmt: *showcpy)
        expropts(*resdecpos)
        extbinint(*yes);

// variabile di tipo SQL che rappresenta un puntatore al result set
dcl-s MyResult sqltype(result_set_locator);

// output lettura cursore
dcl-ds EMPLOYEE ext template end-ds;
dcl-ds DEPARTMENT ext template end-ds;
dcl-ds dsOut qualified inz;
  EMPNO like(EMPNO);
  LASTNAME like(LASTNAME);
  WORKDEPT like(WORKDEPT);
  DEPTNAME like(DEPTNAME);
end-ds;

// esecuzione stored procedure
exec sql
  call RESSET2();

// associazione tra la variabile result set e la stored procedure
exec sql
  associate result set locator(:MyResult)
    with procedure RESSET2;

// associa la variabile result set a un cursore
exec sql
  allocate cResult cursor for
    result set :MyResult;

dow *on;
  exec sql
    fetch next from cResult
      into :dsOut;

  if sqlcode = 100;
    leave;
  endif;

  // ... elaborazione record...
  // ... bla bla bla ...

enddo;

exec sql
  close cResult;

*inlr = *on;
return;


/* ESEMPIO PROGRAMMA RPG USATO IN STORED PROCEDURE
   PER RESTITUIRE UN RESULT SET

   (c) 2022 MarkOneTools - www. markonetools.it */

create or replace procedure RESSET3(Department char(3))
  dynamic result sets 1
  language rpgle
  specific RESSET3
  external name RESSET3
  parameter style general;

label on procedure RESSET3 is 'Result set - esempio da cursore con parm';


**free

// ESEMPIO PROGRAMMA RPG USATO IN STORED PROCEDURE
//  PER RESTITUIRE UN RESULT SET PARTENDO DA UN CURSORE
//  (c) 2022 MarkOneTools - www. markonetools.it
ctl-opt copyright('MarkOneTools')
        decedit('0,')
        indent(' ')
        option(*nodebugio: *srcstmt: *showcpy)
        expropts(*resdecpos)
        extbinint(*yes);

dcl-pr RESSET3 extpgm('RESSET3');
 *n char(3);
end-pr;
dcl-pi RESSET3;
 pmDept char(3);
end-pi;

exec sql
  declare cEmp cursor for
    select EMPNO, LASTNAME, WORKDEPT, DEPTNAME
      from EMPLOYEE
        inner join DEPARTMENT
          on WORKDEPT = DEPTNO
      where ADMRDEPT = :pmDept
      order by WORKDEPT, LASTNAME;

exec sql
  open cEmp;

exec sql
  set result sets cursor cEmp;

*inlr = *on;
return;


**free

// ESEMPIO UTILIZZO STORED PROCEDURE CHE RESTITUISCE UN RESULT SET

// (c) 2022 MarkOneTools - www.markonetools.it

ctl-opt copyright('MarkOneTools')
        decedit('0,')
        indent(' ')
        option(*nodebugio: *srcstmt: *showcpy)
        expropts(*resdecpos)
        extbinint(*yes);

dcl-pr RESSET3T extpgm('RESSET3T');
 *n char(3);
end-pr;
dcl-pi RESSET3T;
 pmDept char(3);
end-pi;

// variabile di tipo SQL che rappresenta un puntatore al result set
dcl-s MyResult sqltype(result_set_locator);

// output lettura cursore
dcl-ds EMPLOYEE ext template end-ds;
dcl-ds DEPARTMENT ext template end-ds;
dcl-ds dsOut qualified inz;
  EMPNO like(EMPNO);
  LASTNAME like(LASTNAME);
  WORKDEPT like(WORKDEPT);
  DEPTNAME like(DEPTNAME);
end-ds;

if %parms() < 1;
  *inlr = *on;
  return;
endif;

// esecuzione stored procedure
exec sql
  call RESSET3(:pmDept);

// associazione tra la variabile result set e la stored procedure
exec sql
  associate result set locator(:MyResult)
    with procedure RESSET3;

// associa la variabile result set a un cursore
exec sql
  allocate cResult cursor for
    result set :MyResult;

dow *on;
  exec sql
    fetch next from cResult
      into :dsOut;

  if sqlcode = 100;
    leave;
  endif;

  // ... elaborazione record...
  // ... bla bla bla ...

enddo;

exec sql
  close cResult;

*inlr = *on;
return;

